# TP1 : Équations Différentielles du 1er ordre

Allumez l'ordinateur si nécessaire, connectez-vous puis lancez `Scilab`.

## Warm up

Dans la console `Scilab`, tapez `x=1:10` puis `Return` (dans la suite on ne précise plus `Return`). Tapez ensuite `x=1:0.5:10`. Tapez à nouveau `x=1:10` puis `x.*x`. Qu'observez-vous ?

Tapez `x=0:0.1:10; y=exp(-x); plot2d(x,y,1); z=sin(x); plot2d(x,z,2);`

À vous ! Tracez le graphe de la fonction $x \longmapsto x^2e^{-x}$ sur l'intervalle $[0, 10]$ puis celui de la fonction $x \longmapsto \sin(3x) - \cos(2x)$ sur $[0, 3\pi]$ ($\pi$=%pi dans `Scilab`).

## Résolution des équations différentielles du 1er ordre

`Scilab` sait résoudre de manière numérique les équations différentielles du type

$$y'(x) = F(x, y(x)), \qquad y(x_0) = y_0. \tag{1}$$

Par exemple, l'équation différentielle $y'(x) + 2y(x) = 5\cos(x)$ se réécrit

$$y'(x) = -2y(x) + 5\cos(x) \qquad F(x, y) = -2y + 5\cos(x). \tag{2}$$

Déterminez la solution de $y'(x) + 2y(x) = 5\cos(x)$ vérifiant $y(0) = 0$ ... Vous obtenez

$$y(x) = -2e^{-2x} + 2\cos(x) + \sin(x).$$

La commande `Scilab` pour résoudre (1) est `ode` ; nous allons l'utiliser pour vérifiez que le calcul précédent est juste. Ouvrez l'éditeur de `Scilab`. Tapez les instructions

```
function ydot=F(x,y)
    ydot=-2*y+5*cos(x);
endfunction
y0=0; x0=0; xmax=10; dx=0.1;
x=x0:dx:xmax;
y=ode(y0,x0,x,F);
clf;
plot2d(x,y,2);
```

Sauvez le fichier sous le nom `exemple1.sce` puis exécutez-le en cliquant sur le triangle. Vous obtenez le graphe de la solution calculée par `Scilab`. Modifiez le programme pour qu'il trace le graphe de la solution exacte de l'équation : rajoutez

```
z=-2*exp(-2*x)+2*cos(x)+sin(x);
plot2d(x,z,-1);
```

Faites un fichier `exemple2.sce` pour résoudre numériquement l'équation différentielle

$$\forall x > 0, \quad x\,y'(x) - y(x) = \ln(x), \qquad y(1) = 1,$$

dont la solution exacte est $y(x) = 2x - \ln(x) - 1$.

## Chute d'un corps

Lorsqu'on néglige les frottements dus à la résistance de l'air, la vitesse d'un corps en chute libre, sans vitesse initiale, est $v(t) = gt$. Ce modèle n'est pas réaliste puisque dans ce cas $\lim_{t\to\infty} v(t) = +\infty$. Un premier modèle considère que les frottements sont proportionnels à la vitesse qui vérifie alors l'équation différentielle :

$$v'(t) = g - \frac{k_l}{m} v(t), \quad t \geq 0.$$

De manière plus réaliste, les frottements sont considérés proportionnels au carré de la vitesse :

$$v'(t) = g - \frac{k_q}{m} v^2(t), \quad t \geq 0.$$

Écrire un programme `Scilab` (`exemple3.sce`) permettant de tracer le graphe de la vitesse $v(t)$ sur $[0, t_{\max}]$ dans le cas de frottements linéaires puis quadratiques. On prendra $t_{\max} = 40$, $m = 80$, $k_l = 14.82$, $k_q = 0.28$ et $g = 9.81$. On commencera par $v(0) = 0$ puis on fera varier $v(0)$. Qu'observez-vous pour le régime permanent ?

## Système différentiel du 1$^{\text{er}}$ ordre

La première partie porte sur l'exercice 5 traitant du refroidissement d'une « pièce » en la plongeant dans un bain d'huile. Le bilan thermique donne ($x$ température du bain d'huile, $y$ température de la pièce) :

$$x'(t) = -0.01\left[x(t) - y(t)\right], \qquad y'(t) = 0.02\left[x(t) - y(t)\right], \qquad x(0) = 30, \quad y(0) = 300.$$

Nous allons résoudre directement ce système différentiel du 1$^{\text{er}}$ ordre.

Créez le fichier `example4.sce` avec les instructions suivantes :

```
function[f]=bdh(t,u)
 f(1)=-0.01*(u(1)-u(2))
 f(2)=0.02*(u(1)-u(2))
endfunction
t0=0; dt=0.1; tmax=200;
u0=[30 ; 300];
t=t0:dt:tmax;
[u]=ode(u0,t0,t,bdh);
clf;
plot2d(t,u(1,:)',2);
plot2d(t,u(2,:)',5);
```

Modifiez le fichier pour représenter dans le plan l'évolution du point $(x, y)$ :

```
xset('window',1);
plot2d(u(1,:)',u(2,:)',5);
```

**Système de Lotka-Volterra.** Il s'agit d'un modèle pour la compétition entre deux espèces : sardines et requins à l'origine. $x$ représente le nombre de proies, $y$ le nombre de prédateurs. L'évolution du système est donné par

$$x'(t) = 3x(t) - x(t)y(t), \quad y'(t) = x(t)y(t) - 4y(t), \quad x(0) = y(0) = 10.$$

Résoudre numériquement ce système différentiel et tracer les graphes des fonctions $x$ et $y$ sur $[0, 15]$ ainsi que l'évolution du point $(x, y)$.